LINGO

LINGO软件及应用习题解答

孙玺菁　司守奎　主编

国防工業出版社
National Defense Industry Press

LINGO 软件及应用
习题解答

孙玺菁　司守奎　主编
邹海滨　周　刚　于方红　编著

国防工业出版社
·北京·

内容简介

本书是国防工业大学出版社出版的《LINGO 软件及应用》一书的配套书籍。本书给出了《LINGO 软件及应用》中全部习题的解答过程和配套程序设计。尤其是第 12 章,针对数学建模竞赛 10 道竞赛题目,参照公开出版的优秀论文给出的解答思路,给出了包括问题分析,模型假设,模型建立和求解,以及对应的程序设计在内的完整解答过程,有些题目还给出了两种不同思路的解答过程。

本书的程序来自于教学实践,有许多经验心得体现在编程的技巧中。这些技巧不仅实用,而且很有特色。书中提供了全部习题的程序,可以直接作为工具箱来使用。

本书可作为开设"数学建模"课程和辅导数学建模竞赛的教师的教材或参考资料,也可作为《LINGO 软件及应用》自学者的参考书,还可供参加数学建模竞赛的本科生和研究生以及科技工作者使用。

图书在版编目(CIP)数据

LINGO 软件及应用习题解答/孙玺菁,司守奎主编.
—北京:国防工业出版社,2018.7
ISBN 978-7-118-11651-9

Ⅰ.①L… Ⅱ.①孙… ②司… Ⅲ.①数学模型-建立模型-应用软件-高等学校-题解 Ⅳ.①O141.4-44

中国版本图书馆 CIP 数据核字(2018)第 158742 号

※

国防工业出版社出版发行
(北京市海淀区紫竹院南路 23 号 邮政编码 100048)
三河市德鑫印刷厂印刷
新华书店经售

*

开本 787×1092 1/16 **印张** 16½ **字数** 414 千字
2018 年 7 月第 1 版第 1 次印刷 **印数** 1—3000 册 **定价** 36.00 元

国防书店:(010)88540777 发行邮购:(010)88540776
发行传真:(010)88540755 发行业务:(010)88540717

前 言

本书是国防工业出版社出版的《LINGO 软件及应用》一书的配套书籍。本书给出了课后所有习题的解答过程和 LINGO 程序设计,第 12 章包含 10 道往年数学建模竞赛本科、专科题目,结合已经公开出版的优秀论文的建模思路,基于 LINGO 软件设计求解程序。

习题是消化领会教材和巩固所学知识的重要环节,本书的出版将为广大读者提供很好的自学参考资料。书中的程序设计思路是作者多年编程经验技巧的总结,也可以为广大读者开阔思路,从而利用 LINGO 软件设计程序求解更多的问题。

对于数学建模的一些综合性题目,本书选择的解答是比较优秀的,但是这类题目的解答是不唯一的。读者应该努力开发自己的想象力和创造力,争取构造有特色的模型。由于编者水平有限,书中难免存在错误或不妥之处,恳请广大读者批评指正。

最后,十分感谢国防工业出版社对本书出版所给予的大力支持,尤其是责任编辑丁福志的热情支持和帮助。

需要本书源程序电子文档的读者,可以通过电子邮件联系索取。E-mail:896369667@QQ. com,sishoukui@163. com。

编者

2018 年 4 月

目　录

第 1 章　LINGO 软件的基本用法习题解答

1.1　用 LINGO 求解下列线性规划：

（1）$\max z=6x_1+2x_2+3x_3+9x_4$,

$$\text{s. t.}\begin{cases}5x_1+6x_2-4x_3-4x_4\leqslant 2,\\3x_1-3x_2+2x_3+8x_4\leqslant 25,\\4x_1+2x_2-x_3+3x_4\leqslant 10,\\x_i\geqslant 0,i=1,2,3,4.\end{cases}$$

（2）$\min z=10x_1+2x_2+x_3+8x_4+6x_5$,

$$\text{s. t.}\begin{cases}x_1+x_3=100,\\x_2+x_4=200,\\x_3+x_5=300,\\x_4+x_5=500,\\x_1+2x_2+x_3+x_4-x_5\geqslant -400,\\2x_1+3x_4+5x_5\geqslant -220.\end{cases}$$

（3）$\max z=8x_1+6x_2+5x_3+9x_4+3x_5$,

$$\text{s. t.}\begin{cases}2x_1+9x_2-x_3-3x_4-x_5\leqslant 20,\\x_1-3x_2+2x_3+6x_4+x_5\leqslant 30,\\x_1+2x_2-x_3+x_4-2x_5\leqslant 10,\\a_i\leqslant x_i\leqslant b_i,i=1,2,3,4,5.\end{cases}$$

其中$[a_1,a_2,a_3,a_4,a_5]=-[10,50,15,20,30]$，$[b_1,b_2,b_3,b_4,b_5]=[20,50,60,30,10]$。

解　（1）LINGO 程序如下：

```
model:
sets:
row/1..3/:b;
col/1..4/:c,x;
link(row,col):a;
endsets
data:
c=6 2 3 9;
a=5 6 -4 -4   3 -3 2 8   4 2 -1 3;
b=2 25 10;
enddata
max=@sum(col:c*x);
@for(row(i):@sum(col(j):a(i,j)*x(j))<b(i));
```

```
end
```

求得最优解为 $x_1=0, x_2=45, x_3=80, x_4=0$，目标函数的最优值 $z=330$。

(2) LINGO 程序如下：

```
model:
sets:
row1/1..4/:b1;
row2/1 2/:b2;
col/1..5/:c,x;
link1(row1,col):a1;
link2(row2,col):a2;
endsets
data:
c=10 2 1 8 6;
a1=1 0 1 0 0   0 1 0 1 0   0 0 1 0 1   0 0 0 1 1;
a2=1 2 1 1 -1   2 0 0 3;
b1=100 200 300 500;
b2=-400 -220;
enddata
min=@sum(col:c*x);
@for(row1(i):@sum(col(j):a1(i,j)*x(j))=b1(i));
@for(row2(i):@sum(col(j):a2(i,j)*x(j))>b2(i));
@for(col:@free(x));
end
```

求得最优解 $x_1=-530, x_2=-630, x_3=630, x_4=830, x_5=-330$，目标函数的最优值 $z=-1270$。

(3) LINGO 程序如下：

```
model:
sets:
row/1..3/:b;
col/1..5/:c,x,L,U;
link(row,col):a;
endsets
data:
c=8,6,5,9,3;
a=2,9,-1,-3,-1,   1,-3,2,6,1,   1,2,-1,1,-2;
b=20,30,10;
L=-10,-50,-15,-20,-30;
U=20,50,60,30,10;
enddata
min=@sum(col:c*x);
@for(row(i):@sum(col(j):a(i,j)*x(j))<b(i));
@for(col:@bnd(L,x,U));
end
```

求得最优解 $x_1=-10,x_2=-50,x_3=-15,x_4=-20,x_5=-30$,目标函数的最优值 $z=-725$。

1.2　在图 1.1 所示的双杆系统中,已知杆 1 重 $G_1=300$ 牛顿,长 $L_1=2$ 米,与水平方向的夹角 $\theta_1=\pi/6$,杆 2 重 $G_2=200$ 牛顿,长 $L_2=\sqrt{2}$ 米,与水平方向的夹角 $\theta_2=\pi/4$。三个铰接点 A,B,C 所在平面垂直于水平面。求杆 1、杆 2 在铰接点处所受到的力。



图 1.1　双杆系统

解　假设两杆都是均匀的,在铰接点处的受力情况如图 1.2 所示。记 $\theta_1=\pi/6,\theta_2=\pi/4$。对于杆 1,水平方向受到的合力为零,故

$$N_1=N_3,$$

竖直方向受到的合力为零,故

$$N_2+N_4=G_1.$$

以点 A 为支点的合力矩为零,故

$$(L_1\sin\theta_1)N_3+(L_1\cos\theta_1)N_4=\left(\frac{1}{2}L_1\cos\theta_1\right)G_1.$$



图 1.2　两杆受力情况

对于杆 2,类似地有

$$N_5=N_7,N_6=N_8+G_2,(L_2\sin\theta_2)N_7=(L_2\cos\theta_2)N_8+\left(\frac{1}{2}L_2\cos\theta_2\right)G_2.$$

此外还有

$$N_3=N_7,\ N_4=N_8.$$

于是将上述 8 个公式联立起来得到关于 $N_1,N_2,\cdots,N_8$ 的线性方程组:

$$\begin{cases}N_1-N_3=0,\\ N_2+N_4=G_1,\\ (L_1\sin\theta_1)N_3+(L_1\cos\theta_1)N_4=\left(\dfrac{1}{2}L_1\cos\theta_1\right)G_1,\\ N_5-N_7=0,\\ N_6-N_8=G_2,\\ (L_2\sin\theta_2)N_7-(L_2\cos\theta_2)N_8=\left(\dfrac{1}{2}L_2\cos\theta_2\right)G_2,\\ N_3-N_7=0,\\ N_4-N_8=0.\end{cases}$$

解上面的线性方程组,求得

$N_1=158.4936, N_2=241.5064, N_3=158.4936, N_4=58.49365,$

$N_5=158.4936, N_6=258.4936, N_7=158.4936, N_8=58.49365.$

计算的 LINGO 程序如下：

```
model:
G1=300; L1=2; theta1=@pi()/6; G2=200; L2=@sqrt(2); theta2=@pi()/4;
N1-N3=0;
N2+N4=G1;
L1*@sin(theta1)*N3+L1*@cos(theta1)*N4=1/2*L1*@cos(theta1)*G1;
N5-N7=0;
N6-N8=G2;
L2*@sin(theta2)*N7-L2*@cos(theta2)*N8=1/2*L2*@cos(theta2)*G2;
N3-N7=0;
N4-N8=0;
end
```

注 1.1 上述 LINGO 程序必须在 LINGO12 下运行，因为调用了函数@pi()。

1.3 求解线性方程组

$$\begin{cases} 4x_1+x_2 = 1, \\ x_1+4x_2+x_3 = 2, \\ x_2+4x_3+x_4 = 3, \\ \quad\ddots \\ x_{998}+4x_{999}+x_{1000}=999, \\ x_{999}+4x_{1000}=1000. \end{cases}$$

解 这里不给出用 LINGO 软件求解的结果。

计算的 LINGO 程序如下：

```
model:
sets:
num/1..1000/:x;
endsets
4*x(1)+x(2)=1;
@for(num(i)|i#gt#1 #and# i#lt#1000:x(i-1)+4*x(i)+x(i+1)=i);
x(999)+4*x(1000)=1000;
@for(num:@free(x));
end
```

1.4 设 3 阶对称阵 $\boldsymbol{A}$ 的特征值为 $\lambda_1=1, \lambda_2=-1, \lambda_3=0$；对应 λ_1, λ_2 的特征向量依次为

$$\boldsymbol{p}_1=\begin{bmatrix}1\\2\\2\end{bmatrix}, \boldsymbol{p}_2=\begin{bmatrix}2\\1\\-2\end{bmatrix},$$

求 $\boldsymbol{A}$。

解 先求 0 对应的特征向量 $\boldsymbol{p}_3$，由实对称矩阵的性质知，$\boldsymbol{p}_3$ 与 $\boldsymbol{p}_1, \boldsymbol{p}_2$ 正交，于是 $\boldsymbol{p}_3$ 可取为方程

$$\begin{bmatrix} \boldsymbol{p}_1^{\mathrm{T}} \\ \boldsymbol{p}_2^{\mathrm{T}} \end{bmatrix} \boldsymbol{x} = 0$$

的任意非零解向量。

由 LINGO 软件求得$\boldsymbol{p}_3^{\mathrm{T}} = [2,-2,1]^{\mathrm{T}}$。

则所求的实对称矩阵 $\boldsymbol{A} = (a_{ij})_{3\times 3}$ 满足 $a_{ij} = a_{ji}, i,j = 1,2,3$,且

$$\boldsymbol{A}\boldsymbol{p}_1 = \lambda_1 \boldsymbol{p}_1, \boldsymbol{A}\boldsymbol{p}_2 = \lambda_2 \boldsymbol{p}_2, \boldsymbol{A}\boldsymbol{p}_3 = \lambda_3 \boldsymbol{p}_3.$$

解上面的方程组,即可求得矩阵

$$\boldsymbol{A} = \begin{bmatrix} -0.333333 & 0 & 0.666667 \\ 0 & 0.333333 & 0.666667 \\ 0.666667 & 0.666667 & 0 \end{bmatrix}.$$

注 1.2　上面求解结果是数值解,手工求解的精确结果为

$$\boldsymbol{A} = \frac{1}{3} \begin{bmatrix} -1 & 0 & 2 \\ 0 & 1 & 2 \\ 2 & 2 & 0 \end{bmatrix}.$$

计算的 LINGO 程序如下:

```
model:
sets:
num/1..3/:p1,p2,p3,p4;
link(num,num):a;
endsets
data:
p1=1 2 2;
p2=2 1 -2;
enddata
submodel sp3:
@sum(num:p1*p3)=0;
@sum(num:p2*p3)=0;
@sum(num:p3)=1; !为了求非零向量,这里约束所有的分量和为1;
@for(num:@free(p3));
endsubmodel
submodel fangcheng:
@for(num(i):@sum(num(j):a(i,j)*p1(j))=p1(i));
@for(num(i):@sum(num(j):a(i,j)*p2(j))=-p2(i));
@for(num(i):@sum(num(j):a(i,j)*p4(j))=0);
@for(link(i,j)|i#ne#j:a(i,j)=a(j,i));!实对称性要求;
@for(link:@free(a));
endsubmodel
calc:
@solve(sp3); @for(num:p4=p3); !为了传递参数,把 p3 改写成 p4;
@solve(fangcheng);
endcalc
end
```

1.5 已知矩阵 $\boldsymbol{A}=\begin{bmatrix}1&4&5\\4&2&6\\5&6&3\end{bmatrix}$，$\boldsymbol{x}=\begin{bmatrix}x_1\\x_2\\x_3\end{bmatrix}$，求二次型 $f(x_1,x_2,x_3)=\boldsymbol{x}^{\mathrm{T}}\boldsymbol{A}\boldsymbol{x}$ 在单位球面 $x_1^2+x_2^2+x_3^2=1$ 上的最小值。

解 利用 LINGO 软件，求得的最小点为 $x_1=0.3130$，$x_2=0.5774$，$x_3=-0.7541$，求得的最小值为 -3.6687。

计算的 LINGO 程序如下：

```
model:
sets:
num/1..3/:x;
link(num,num):a;
endsets
data:
a=1 4 5 4 2 6 5 6 3;
enddata
min=@sum(link(i,j):a(i,j)*x(i)*x(j));
@sum(num(i):x(i)^2)=1;
@for(num(i):@free(x(i)));
end
```

1.6 求解下列非线性整数规划问题

$$\max z=x_1^2+x_2^2+3x_3^2+4x_4^2+2x_5^2-8x_1-2x_2-3x_3-x_4-2x_5,$$

$$\text{s.t.}\begin{cases}0\leqslant x_i\leqslant 99,\text{且 } x_i \text{ 为整数}, & i=1,\cdots,5,\\ x_1+x_2+x_3+x_4+x_5\leqslant 400,\\ x_1+2x_2+2x_3+x_4+6x_5\leqslant 800,\\ 2x_1+x_2+6x_3\leqslant 200,\\ x_3+x_4+5x_5\leqslant 200.\end{cases}$$

解 利用 LINGO 软件求得的最优解为

$$x_1=50, x_2=99, x_3=0, x_4=99, x_5=20,$$

目标函数的最大值为 $z=51568$。

计算的 LINGO 程序如下：

```
model:
sets:
num/1..5/:c1,c2,x; !c1为目标函数二次项的系数,c2一次项的系数;
row/1..4/:b; !b为约束条件右边的常数项列;
link(row,num):a;
endsets
data:
c1=1 1 3 4 2;
c2=-8 -2 -3 -1 -2;
a=1 1 1 1 1  1 2 2 1 6  2 1 6 0 0  0 0 1 1 5;
```

```
b=400 800 200 200;
enddata
max=@sum(num(j):c1(j)*x(j)^2+c2(j)*x(j));
@for(row(i):@sum(num(j):a(i,j)*x(j))<=b(i));
@for(num(j):@bnd(0,x(j),99); @gin(x(j)));
end
```

1.7　用 LINGO 求方程组

$$\begin{cases} x^2+y^2=2, \\ 2x^2+x+y^2+y=4 \end{cases}$$

的所有实数解。

解　LINGO 程序如下：

```
model:
submodel maincon:              !定义方程子模块;
x^2+y^2=2;
2*x^2+x+y^2+y=4;
endsubmodel
submodel con1:                 !定义附加约束子模块;
@free(x);x<0;
endsubmodel
submodel con2:                 !定义附加约束子模块;
@free(y); y<0;
endsubmodel
submodel con3:                 !定义附加约束子模块;
@free(x); @free(y);
x<0; y<0;
endsubmodel
calc:
@solve(maincon);               !调用子模块;
@ifc(@status()#eq#1:@write('没有可行解.',@newline(2));
@else
@write('所求的解为 x=',x,',y=',y,'.',@newline(2)));
@solve(maincon,con1);
@ifc(@status()#eq#1:@write('没有可行解.',@newline(2));
@else
@write('所求的解为 x=',x,',y=',y,'.',@newline(2)));
@solve(maincon,con2);
@ifc(@status()#eq#1:@write('没有可行解.',@newline(2));
@else
@write('所求的解为 x=',x,',y=',y,'.',@newline(2)));
@solve(maincon,con3);
@ifc(@status()#eq#1:@write('没有可行解.',@newline(2));
@else
```

```
@write('所求的解为 x=',x,',y=',y,'.',@newline(2)));
endcalc
end
```

1.8 求解下列非线性规划：

(1) $\max z=\sum_{i=1}^{100} \sqrt{x_i}$,

$$\text{s.t.}\begin{cases} x_1 \leqslant 10, \\ x_1+2x_2 \leqslant 20, \\ x_1+2x_2+3x_3 \leqslant 30, \\ x_1+2x_2+3x_3+4x_4 \leqslant 40, \\ \sum_{i=1}^{100}(101-i)x_i \leqslant 10000, \\ x_i \geqslant 0, i=1,2,\cdots,100. \end{cases}$$

(2) $\max z=(x_1-1)^2+\sum_{i=1}^{99}(x_i-x_{i+1})^2$,

$$\text{s.t.}\begin{cases} x_1+\sum_{i=2}^{100} x_i^2=200, \\ \sum_{i=1}^{50} x_{2i}^2-\sum_{i=1}^{50} x_{2i-1}^2=10, \\ \left(\sum_{i=1}^{33} x_{3i}\right)\left(\sum_{i=1}^{50} x_{2i}\right) \leqslant 1000, \\ -5 \leqslant x_i \leqslant 5, i=1,2,\cdots,100. \end{cases}$$

解 (1) LINGO 程序如下：

```
model:
sets:
num/1..100/:x;
endsets
max=@sum(num:@sqrt(x));
@for(num(i)|i#le#4:@sum(num(j)|j#le#i:j*x(j))<10*i);
@sum(num(i):(101-i)*x(i))<10000;
end
```

(2) LINGO 程序如下：

```
model:
sets:
num/1..100/:x;
endsets
max=(x(1)-1)^2+@sum(num(i)|i#le#99:(x(i)-x(i+1))^2);
x(1)+@sum(num(i)|i#ge#2:x(i)^2)=200;
@sum(num(i)|i#le#50:x(2*i)^2-x(2*i-1)^2)=10;
@sum(num(i)|i#le#33:x(3*i))*@sum(num(i)|i#le#50:x(2*i))<1000;
```

```
@for(num:@bnd(-5,x,5));
end
```

1.9　已知

$$A=[A_1,A_2,\cdots,A_6]=\begin{bmatrix}89.39 & 86.25 & 108.13 & 106.38 & 62.40 & 47.19\\ 64.3 & 99 & 99.6 & 96 & 96.2 & 79.9\end{bmatrix},$$

$$B=[B_1,B_2,\cdots,B_6]=\begin{bmatrix}25.2 & 28.2 & 29.4 & 26.4 & 27.2 & 25.2\\ 223 & 287 & 317 & 291 & 295 & 222\end{bmatrix}.$$

利用 LINGO 的子模块，对于 $j_0=1,2,\cdots,6$，求解如下 6 个线性规划问题：

$$\max B_{j_0}^{\mathrm{T}}Y,$$

$$\text{s. t.}\begin{cases}A_j^{\mathrm{T}}X-B_j^{\mathrm{T}}Y\geqslant 0, & j=1,2,\cdots,6,\\ A_{j_0}^{\mathrm{T}}X=1, \\ X\geqslant 0,Y\geqslant 0.\end{cases}$$

其中 $X=[x_1,x_2]^{\mathrm{T}}$，$X\geqslant 0$ 表示 $x_1,x_2\geqslant 0$；$Y=[y_1,y_2]^{\mathrm{T}}$，$Y\geqslant 0$ 表示 $y_1,y_2\geqslant 0$.

解　LINGO 程序如下：

```
model:
sets:
dmu/1..6/;
inw/1..2/:Aj0,x;        !x 输入权重;
outw/1..2/:c,y;         !c,y 输出权重;
inv(inw,dmu):A;         !输入数据;
outv(outw,dmu):B;       !输出数据;
endsets
data:
A=89.39   86.25   108.13   106.38   62.40   47.19
  64.3    99      99.6     96       96.2    79.9;
B=25.2    28.2    29.4     26.4     27.2    25.2
  223     287     317      291      295     222;
enddata
submodel mylinprog:
[obj]max=@sum(outw:c*y);
@for(dmu(j):@sum(inw(i):A(i,j)*x(i))-@sum(outw(k):B(k,j)*y(k))>0);
@sum(inw:Aj0*x)=1;
endsubmodel
calc:
@for(dmu(j0):@for(outw(k):c(k)=B(k,j0));@for(inw(i):Aj0(i)=A(i,j0));
@solve(mylinprog);
@write('第',j0,'个评价对象的目标函数值为',@format(obj,'8.6f'),@newline(1));
@write('x=');@writefor(inw(i):'  ',@format(x(i),'7.6f'));
@write(';    y=');@writefor(outw(k):'  ',@format(y(k),'7.6f'));@write(@newline(1)));
endcalc
end
```

第2章 LINGO软件与外部文件的接口习题解答

2.1 假设海军要在全国范围内建立装备器材的配送中心，军事物流配送中心备选点有8个，有15个部队用户。根据以往的历史数据得知运输费率为0.32元/(吨·千米)，运输速度为20千米/小时，当配送时间在55.6小时之内时，能够达到部队用户对时效性的要求。每个部队用户由一个配送中心进行供给。试利用表2.1和表2.2给定的数据，研究如下3个问题：

(1) 建立模型，确定最佳选址位置，在满足时效性要求的前提下，使总的成本最小。

(2) 利用文本文件传递数据，利用LINGO软件求(1)中所建立的模型。

(3) 利用Excel文件传递数据，利用LINGO软件求(1)中所建立的模型。

表2.1 配送中心的固定成本和容量限制

配送中心	1	2	3	4	5	6	7	8
固定成本/万元	2300	2400	2400	3200	2700	3200	2500	2700
容量限制/吨	18600	19600	17100	18900	17000	19100	20500	17200

表2.2 部队用户和配送中心之间的距离和需求量

部队用户	配送中心/千米								需求量/吨
	1	2	3	4	5	6	7	8	
1	390.6	618.5	553	442	113.1	5.2	1217.7	1011	3000
2	370.8	636	440	401.8	25.6	113.1	1172.4	894.5	3100
3	876.3	1098.6	497.6	779.8	903	1003.3	907.2	40.1	2900
4	745.4	1037	305.9	725.7	445.7	531.4	1376.4	768.1	3100
5	144.5	354.6	624.7	238	290.7	269.4	993.2	974	3100
6	200.2	242	691.5	173.4	560	589.7	661.8	855.7	3400
7	235	205.5	801.5	326.6	477	433.6	966.4	1112	3500
8	517	541.5	338.4	219	249.5	335	937.3	701.8	3200
9	542	321	1104	576	896.8	878.4	728.3	1243	3000
10	665	827	427	523.2	725.2	813.8	692.2	284	3100
11	799	855.1	916.5	709.3	1057	1115.5	300	617	3300
12	852.2	798	1083	714.6	1177.4	1216.8	40.8	898.2	3200
13	602	614	820	517.7	899.6	952.7	272.4	727	3300
14	903	1092.5	612.5	790	932.4	1034.9	777	152.3	2900
15	600.7	710	522	448	726.6	811.8	563	426.8	3100

解 (1) 用 $i=1,2,\cdots,15$ 表示部队用户编号，$j=1,2,\cdots,8$ 表示配送中心备选点编号；a_j 表示第 j 个配送中心备选点的固定成本，b_j 表示第 j 个配送中心备选点的容量；c_i 表示第 i 个部队用户的需求量，d_{ij} 表示第 i 个部队用户到第 j 个配送中心备选点的距离。

引进 0-1 变量 i

$$x_j=\begin{cases}1, & \text{第 } i \text{ 个备选点建立配送中心},\\ 0, & \text{第 } i \text{ 个备选点不建配送中心};\end{cases}$$

$$y_{ij}=\begin{cases}1, & \text{第 } j \text{ 个配送中心供应第 } i \text{ 个用户},\\ 0, & \text{第 } j \text{ 个配送中心不供应第 } i \text{ 个用户}.\end{cases}$$

费用包括两部分,建立配送中心的固定成本和配送中心到部队的运费。总的费用

$$z=\sum_{j=1}^{8}a_jx_j+\sum_{i=1}^{15}\sum_{j=1}^{8}0.32c_id_{ij}y_{ij}.$$

约束条件包含 3 类。

① 时效性要求,平均配送时间在 55.6 小时之内,即

$$\frac{d_{ij}}{20}y_{ij}\leqslant 55.6x_j, i=1,2,\cdots,15, j=1,2,\cdots,8.$$

② 对每个部队用户都要由唯一的一个配送中心进行供给,即

$$\sum_{j=1}^{8}y_{ij}=1, i=1,2,\cdots,15.$$

③ 配送中心的容量限制,即

$$\sum_{i=1}^{15}c_iy_{ij}\leqslant b_j, j=1,2,\cdots,8.$$

综上所述,建立如下的线性规划模型:

$$\min z=\sum_{j=1}^{8}a_jx_j+\sum_{i=1}^{15}\sum_{j=1}^{8}0.32c_id_{ij}y_{ij}$$

$$\text{s.t.}\begin{cases}\dfrac{d_{ij}}{20}y_{ij}\leqslant 55.6x_j, & i=1,2,\cdots,15, j=1,2,\cdots,8,\\ \sum\limits_{j=1}^{8}y_{ij}=1, & i=1,2,\cdots,15,\\ \sum\limits_{i=1}^{15}c_iy_{ij}\leqslant b_j, & j=1,2,\cdots,8.\\ x_j=0 \text{ 或 } 1, & j=1,2,\cdots,8,\\ y_{ij}=0 \text{ 或 } 1, & i=1,2,\cdots,15, j=1,2,\cdots,8.\end{cases}$$

(2) 利用文本文件传递数据,利用 LINGO 软件求得需要设置 3 个配送中心,选址方案见表 2.3;总的成本为 7.73748×10^7 元。

表 2.3　最终的选址方案

选址的配送中心	负责的部队用户
1	1,2,5,6,7
3	3,4,8,10,14
7	9,11,12,13,15

计算的 LINGO 程序如下:

```
model:
sets:
```

```
yonghu/1..15/:c;
dian/1..8/:a,b,x;
link(yonghu,dian):d,y;
endsets
data:
a=@file('Ldata21.txt');
b=@file('Ldata21.txt');
c=@file('Ldata21.txt');
d=@file('Ldata21.txt');
enddata
min=10000*@sum(dian:a*x)+@sum(link(i,j):0.32*c(i)*d(i,j)*y(i,j));
@for(link(i,j):d(i,j)*y(i,j)/20<55.6*x(j));
@for(yonghu(i):@sum(dian(j):y(i,j))=1);
@for(dian(j):@sum(yonghu(i):c(i)*y(i,j))<b(j));
@for(dian:@bin(x));
@for(link:@bin(y));
end
```

(3) 利用 Excel 文件传递数据,计算的 LINGO 程序如下:

```
model:
sets:
yonghu/1..15/:c;
dian/1..8/:a,b,x;
link(yonghu,dian):d,y;
endsets
data:
a=@ole('Ldata22.xlsx','B3:I3');
b=@ole('Ldata22.xlsx','B4:I4');
c=@ole('Ldata22.xlsx','J9:J23');
d=@ole('Ldata22.xlsx','B9:I23');
@ole('Ldata22.xlsx','A31:H45')=y;   !计算结果输出到 Excel 文件中;
enddata
min=10000*@sum(dian:a*x)+@sum(link(i,j):0.32*c(i)*d(i,j)*y(i,j));
@for(link(i,j):d(i,j)*y(i,j)/20<55.6*x(j));
@for(yonghu(i):@sum(dian(j):y(i,j))=1);
@for(dian(j):@sum(yonghu(i):c(i)*y(i,j))<b(j));
@for(dian:@bin(x));
@for(link:@bin(y));
end
```

2.2 求解数学规划问题

$$\max \sum_{i=1}^{200} |x_i|,$$

$$\text{s.t.} \begin{cases} \boldsymbol{Ax} \leqslant 1_{50\times 1}, \\ -1 \leqslant x_i \leqslant 1, i=1,2,\cdots,200. \end{cases}$$